
1 Introduction
MapReduce is a general technique for massively parallel programming developed by 
Google.  It takes its inspiration from ideas in functional programming, but has moved away 
from that paradigm to a more imperative approach.
This  note demonstrates that MapReduce can be expressed naturally, using functional 
programming techniques, as a form of monad.   All code examples are written in Haskell.  
Code files can be downloaded from the JP Embedded Solutions blog.  

2 MapReduce
2.1 Description
MapReduce consists  of two stages named map and reduce.  In map, the following 
happens:
1. A central unit takes data in the form of key / value pairs  and divides it into a number of 

chunks, which it distributes to:
2. One or more mappers, each of which transforms its input key / value pairs  into an 

output list of key / value pairs.  
In reduce, the following happens:
1. A central unit sorts the output of map on key value, dividing it into chunks, one for each 

unique key value in the set, and distributes the chunks to:
2. One instance for each chunk of a reducer, which transforms its input key / value pairs 

into an output list of key / value pairs.
This process is repeated until the desired result is achieved.
Note that:

• The mapper and reducer can be many to many functions, so one input pair may 
correspond to no or more output pairs.  

• The type of the output from these functions need not be the same as that of the input.  

• In practice there is an intermediate stage called partitioning, which is essentially a 
second reduce stage, so discussion of reduce applies equally to it.

2.2 Comments
It should be observed that MapReduce is simply a form of the standard scatter-gather 
paradigm for highly parallel problems, which is a series of stages, in each of which a 
central controller takes the data, chops it into chunks, hands off each chunk to a 
processing node, gathers the results from each of the nodes, and concatenates them 
ready for the next stage.

Also note that the only distinction between map and reduce lies in how the initial dataset is 
chopped up.  In map no criterion is specified (it could be random), whereas in reduce it is 
base on key value.

3 Generalising MapReduce
3.1 Theory
3.1.1 Transformer functions
It makes sense to treat mappers and reducers similarly, so call them generically 
transformers.  Transformers map a list of values to a list of value / key pairs, and so are 
functions of type:
[s] -> [(s',a)]

First consider the reduce stage.  It collects data from the map stage and then, partitions it 
into a number of sets, one per key value, and applies the reducer separately to each set.  
So, for each key value k it wraps a reducer by stripping out the pairs  that correspond to k 
and then applying the reducer to them:
reduceF :: (Eq b) => b -> ([s'] -> [(s'',c)]) -> ([(s',b)] -> [(s'',c)])
reduceF k reducer = (\ss -> reducer $ partition k ss)
        where
        partition k ss = fst <$> filter (\(_,k') -> k == k') ss

Note the signature of the wrapper function:
b -> ([s'] -> [(s'', c)]) -> ([(s', b)] -> [(s'', c)])

A moment’s thought shows that the map stage takes the same form.

3.1.2 Generalising transformers
There is no reason why reduceF should be a wrapper around a transformer.  Much is to be 
gained from allowing them to be more general functions with the same signature, 
absorbing the transformer into the body of the function.   Following this approach,  
transformers generalise to functions of type:
a -> ([(s,a)] -> [(s',b)])

This  is highly suggestive of monadic composition.  The only difference from a standard 
monad is that is that the processing step changes the type of the data and the key, and so 
the rule for monadic composition generalises from:
m s a -> ( a -> m s b ) -> m s b
to:
m s a s' b -> ( b -> m s' b s'' c ) -> m s a s'' c

(It is possible to abstract further by treating the key / value pair as a single type, in which 
case the problem reduces to a generalised Haskell State monad.  However, in that case 
monadic composition is opaque, whereas this  definition admits a very natural parallelism.  
Therefore it is retained in spite of the undoubted clumsiness of having to carry around four 
type parameters.)

3.1.3 The monadic approach
The goal is then for a single round of generalised MapReduce to be expressible as:
state >>= mapF >>= reduceF

where mapF and reduceF are the generalised transformers  and state is  a generalised 
monad containing the initial state.  More generally, if mapper and reducer are traditional 
transformers, then there should be a function 
wrapMR :: (Eq a) => ([s] -> [(s',b)]) -> (a -> m s a s' b)

that wraps a transformer in the monad such that traditional MapReduce can be expressed 
as:
. . . >>= wrapMR mapper >>= wrapMR reducer >>= . . .

3.2 Implementation

3.2.1 Definitions
The generalised monad is therefore:
class Monad' m where
        return :: a -> m s x s a
        (>>=)  :: (Eq b) => m s a s' b 
                -> ( b -> m s' b s'' c ) -> m s a s'' c
The generalised transformers are described by a type which is an instance of Monad':
newtype MapReduce s a s' b = MR { runMR :: ([(s,a)] -> [(s',b)]) } 

3.2.2 The monadic operations
The monadic return operator is fairly trivial:
retMR :: a -> MapReduce s x s a
retMR k = MR (\ss -> [(s,k) | s <- fst <$> ss])

Monadic composition is the critical part of the design:
bindMR :: (Eq b,NFData s'',NFData c) => MapReduce s a s' b 
        -> (b -> MapReduce s' b s'' c) -> MapReduce s a s'' c
bindMR f g = MR (\s ->
        let
                fs = runMR f s
                gs = map g $ nub $ snd <$> fs
        in
        concat $ map (\g' -> runMR g' fs) gs)

Note that the set of key values is extracted from the output of f and used, with the 
assistance of g, to generate a set of transformers, one per key value.

Each of these functions is given the whole data set output by f.  It is left to the functions 
themselves to determine which parts of the data to process.  This enables strategies  like 
MapReduce where one function is applied to sets based on key values, strategies where 
varying functions are applied to all of the data, or a mix of the two.
The crucial factor for making this  code parallel is the map in the final line.  In the full listing 
below, this is made explicitly parallel, so each transformer function runs in its own thread.

3.2.3 Relation to standard MapReduce
A standard transformer can be wrapped to produce a monad.  Note that this  approach 
makes use of implicit state to carry out the stripping out of all data values with a given key 
value.
wrapMR :: (Eq a) => ([s] -> [(s',b)]) -> (a -> MapReduce s a s' b)
wrapMR f = (\k -> MR (g k))

        where
        g k ss = f $ fst <$> filter (\s -> k == snd s) ss

3.2.4 Limitations
As it stands, in a parallel implementation, monadic composition sends the entire dataset to 
every thread of execution.  This  may be desirable in cases where generalised MapReduce 
is  being used, but for standard MapReduce with large datasets on many processors, this 
may be an issue.  Possible approaches involve use of shared memory / disk to store the 
datasets, or a more carefully engineered implementation of wrapMR.

4 Code
4.1 The generalised MapReduce library
This  is an implementation of generalised MapReduce that is multithreaded and concurrent 
on multi-processor system with a single OS image.  The library consists of three modules.  

4.1.1 Hashable types module
This  defines a class Hashable of types that can be hashed to an Int.  This is  used by the 
MapReduce library to distribute data items randomly across multiple transformers. 
{-# LANGUAGE TypeSynonymInstances #-}

module Hashable (Hashable, hash) where

import qualified Crypto.Hash.MD5 as H
import Data.Char (ord)
import Data.ByteString (ByteString,pack,unpack,singleton)
import Data.Bits (xor,(.&.))

-- class definition; any type that can convert to ByteString is hashable

class Hashable s where
        conv :: s -> ByteString
        
-- the hash function

hash :: (Hashable s) => s -> Int
hash s = sum $ map fromIntegral (unpack h)
        where
        h = H.hash $ conv s        

{- INSTANCES OF THE CLASS -}
        
-- make String Hashable

instance Hashable String where
        conv s = pack $ map (fromIntegral . ord) s

-- make Integer Hashable
        
instance Hashable Integer where
        conv 0 = singleton 0
        conv x = pack . map fromIntegral $ conv' x

                where
                conv' 0 = []
                conv' y = (y .&. 255):conv' (y `div` 256)

-- make ByteString Hashable

instance Hashable ByteString where
        conv = id                          

4.1.2 parallel map module
This  defines  a parallel version of the standard Prelude function map.  This version is a 
simple multithreaded implementation using standard libraries, but a concurrent version 
could easily be substituted.
module ParallelMap (map) where

import Control.Parallel.Strategies (parMap, rdeepseq)
import Control.DeepSeq (NFData)
import Prelude hiding (map)

map :: (NFData b) => (a -> b) -> [a] -> [b]
map = parMap rdeepseq

4.1.3 MapReduce module
This  defines  the type MapReduce, the generalised Monad, and the support functions that 
allow programs to compose transformations, run generalised MapReduce jobs and wrap 
traditional transformers to fit them within the framework.
module MapReduce (MapReduce, Hashable, return, (>>=), runMapReduce, 
        distributeMR, wrapMR) where

import Data.List (nub)
import Control.Applicative ((<$>))
import Control.Monad (liftM)
import Control.DeepSeq (NFData)
import IO
import Prelude hiding (return,(>>=))
import Hashable (Hashable,hash)
import qualified ParallelMap as P

{-- TYPE DEFINITIONS --}

-- the generalised Monad

class Monad' m where
        return :: a -> m s x s a
        (>>=)  :: (Eq b,NFData s'',NFData c) => m s a s' b 
                -> ( b -> m s' b s'' c ) -> m s a s'' c

-- the generalised MapReduce data type

newtype MapReduce s a s' b = MR { runMR :: ([(s,a)] -> [(s',b)]) }

-- the generalised MapReduce type as an instance of Monad'

instance Monad' MapReduce where
        return = retMR
        (>>=)  = bindMR
        
{-- definition of the Monad' operations for MapReduce --}

-- the return function

retMR :: a -> MapReduce s x s a
retMR k = MR (\ss -> [(s,k) | s <- fst <$> ss])

-- the monadic composition function
       
bindMR :: (Eq b,NFData s'',NFData c) => MapReduce s a s' b 
        -> (b -> MapReduce s' b s'' c) -> MapReduce s a s'' c
bindMR f g = MR (\s ->
        let
                fs = runMR f s
                gs = map g $ nub $ snd <$> fs
        in
        concat $ P.map (\g' -> runMR g' fs) gs)         

{-- utility functions --}

-- execute a MapReduce Monad' given specified initial state

runMapReduce :: MapReduce s () s' b -> [s] -> [(s',b)]
runMapReduce m ss = (runMR m) [(s,()) | s <- ss]

-- distribute key values of a data set using a hash
       
distributeMR :: (Hashable s) => MapReduce s () s Int
distributeMR = MR (\ss -> [(s,hash s) | s <- fst <$> ss])                

-- wrap a traditional mapper / reducer in a monadic function

wrapMR :: (Eq a) => ([s] -> [(s',b)]) -> (a -> MapReduce s a s' b)
wrapMR f = (\k -> MR (g k))
        where
        g k ss = f $ fst <$> filter (\s -> k == snd s) ss        

4.2 An application
This  is the complete code for the classic MapReduce word-count program.  Note that the 
MapReduce part of the application (mapReduce) requires only two lines of active code.  
Also note the use of distributeMR to spread the initial data across a number of mappers.
module Main where

import MapReduce
import IO
import System.Environment (getArgs)
import Control.Monad (liftM)
import Prelude hiding (return,(>>=))

import qualified Prelude as P

main::IO()
main = do
        args <- getArgs
        state <- getLines (head args)
        let res = mapReduce state
        putStrLn $ show res

-- perform MapReduce

mapReduce :: [String] -> [(String,Int)]
mapReduce state = runMapReduce mr state
        where
        mr = distributeMR >>= wrapMR mapper >>= wrapMR reducer 

-- transformers

mapper :: [String] -> [(String,String)]
mapper [] = []
mapper (x:xs) = parse x ++ mapper xs
        where
        parse x = map (\w -> (w,w)) $ words x

reducer :: [String] -> [(String,Int)]
reducer [] = []
reducer xs = [(head xs,length xs)]

-- get input

getLines :: FilePath -> IO [String]
getLines file = do
        h <- openFile file ReadMode
        text <- hGetContents h
        P.return (lines text) 

Finally, the cabal file to build the application is:
name:           MapReduce
version:        0.1
cabal-version:  >= 1.2
Build-Type:     Simple

executable MapReduce
  Build-Depends:   base >= 3 && < 5, 
                   cryptohash >= 0.7,
                   haskell98 >= 1.0.1.1,
                   deepseq >= 1.1.0.0,
                   parallel >= 2.2.0.0,
                   bytestring >=0.9.1.9 
  hs-source-dirs:  src
  main-is:         Main.hs
  other-modules:   MapReduce, Hashable, ParallelMap
  GHC-options:     -threaded

